Lost Messages

by Brian Long, extracted from The Delphi Clinic in Issue 20 of The Delphi Magazine (http://www.itecuk.com/delmag)

I am trying to trap various keystrokes on components on my forms using their OnKeyDown event handlers. There is no problem doing this with a TEdit, but with a TCheckBox and also some other components I do not seem to be able to trap the arrow keys. Why is this and how do I get around the problem?

The normal behaviour for most controls on a form when the cursor keys is pressed is to lose focus. The keystroke is dealt with by the form and the next or previous control in the tab order is given the focus, depending which key was pressed. A TEdit is a wrapper around a Windows edit control. That control specifically traps the up and down keys in order to move the cursor around the edit contents when they are pressed. In order for you to match this functionality you need to know a little about how the VCL deals with keystrokes.

In proper Windows applications, dialogs support keystrokes and the tab key for switching the focus around controls - normal windows do not. Since normal non-dialog windows seem to get the rough end of the deal and since the VCL does not use proper dialog resources, it chooses to step in and emulate the dialog functionality as closely as it can. This means taking the tab and cursor keystrokes out of normal processing, and dealing with them individually.

There seem to be at least four ways of overriding this behaviour. For the Windows hackers, the obvious approach of trapping the wm_KeyDown message in a Windows message handler fails as badly as the event handler, as does overriding the WndProc method and trying to trap wm_KeyDown. The wm_KeyDown message simply does not make it to the control.

Having perused the VCL source, it seems that the message flow goes something like this. Any message that comes into a program via a message queue (that is, one sent via the PostMessage API instead of SendMessage) enters via the main message processing loop in TApplication.ProcessMessage (which is what eventually gets executed when you call Application.ProcessMessages). The general idea of such a loop is that all messages destined for any window in the application enter it and are then dispatched to their rightful destinations by more API calls.

However before any dispatching goes ahead, several other TApplication methods get a chance to fool around with the message and see if there is something extra special that needs to be done with it. They then have the option of stopping the message from being dispatched on its normal route.

The routine that takes a keystroke message is TApplication.IsKeyMsg in the Forms unit. What this function does is to immediately send another message to the window that was the target of the original message (the checkbox in our discussion) using the Windows SendMessage API. This API forces the message to be processed before execution continues in the calling routine. The message that gets sent is an internal VCL component notification message, whose constants are defined in the Controls unit.

Component notifications are always sent from parent type windows to child type windows. In this case, the application window sends it to a checkbox window. To make the component notification message it adds cn_Base ($2000) onto the original message (which means that it will not get confused with any normal Windows messages). If the message was wm_KeyDown it gets turned into cn_KeyDown.

So the checkbox gets sent a cn_KeyDown message. The idea is to let the VCL do its own special pre-processing of certain messages before letting (and sometimes without letting) the normal message handlers into the party. So what happens when the checkbox receives the cn_KeyDown message? Over to the Controls unit to the TWinControl.CNKeyDown message handler.

The first thing this does is to see if the control wants to deal with the key that was pressed by sending it a component message. Where component notification messages start with a cn_ prefix, more general internal component messages start with cm_. You can find a list of all these messages in the help - look up component messages and component notifications. The control sends itself a cm_WantSpecialKey message to see if it is interested in this particular keystroke. If the control responds positively (by returning non-zero) then CNKeyDown exits and lets the normal message handlers proceed. However, hardly any components (with the exception of grid controls) actually respond to this message and so the default result of zero causes us to move onto the next stage.

The code now sends a normal wm_GetDlgCode Windows message to the control to find out in a more honest, above board, regular API-type way if the control wants to deal with the keystroke, providing it is one that would normally be dealt with automatically by dialogs. If it does, then fine - all is normal. However if it does not, the form is then given the chance to act like a dialog by being sent the message details under the guise of a cm_DialogKey message. If it gets to this stage, the keystroke is swallowed and you will not see it in the normal event handlers.

So to alter this behaviour we have the following options:

Create a component descended from TCheckBox and:

· write a cn_KeyDown message handler to avoid this lengthy, winding path if the key is a cursor key

· write a cm_WantSpecialKey message handler and indicate that arrow keys are wanted by the control

· write a wm_GetDlgCode message handler and indicate that arrow keys will be processed by the control

or write a cm_DialogKey message handler in the form class which ignores arrow keys for specified controls.

Four such methods are given in Listings 1 to 4 and there are four projects supplied on the disk, each of which offers a different solution (TESTCHK1.DPR, TESTCHK2.DPR, TESTCHK3.DPR and TESTCHK4.DPR). Note that three of these make use of different components, so they will need to be installed. This can be done by installing the NEWCHKS.PAS unit.

The net result of all this is that all four projects have OnKeyDown event handlers that trap up and down cursor presses for a checkbox. This could be as an OnKeyDown handler for the checkbox component or as an OnKeyDown handler for the form whose KeyPreview is set to True. The supplied projects take the former option.

The same approach would apply equally to trapping the Tab key. In most cases you would substitute vk_Tab for the currently sought after key. In the wm_GetDlgCode message handler you would add in the dlgc_WantTab mask.

Listing 1

TNewCheck1 = class(TCheckBox)

public

  procedure CNKeyDown(var Msg: TWMKeyDown);

    message cn_KeyDown;

end;

...

procedure TNewCheck1.CNKeyDown(var Msg: TWMKeyDown);

begin

  with Msg do

    if CharCode in [vk_Up, vk_Down] then

      { Return 0 to stop the keystrokes being "absorbed" }

      Result := 0

    else

      inherited

end;

Listing 2

TNewCheck2 = class(TCheckBox)

public

  procedure CMWantSpecialKey(var Msg: TCMWantSpecialKey);

    message cm_WantSpecialKey;

end;

...

procedure TNewCheck2.CMWantSpecialKey(var Msg: TCMWantSpecialKey);

begin

  inherited;

  with Msg do

    if CharCode in [vk_Up, vk_Down] then

      { Return 1 to stop the keystrokes being "absorbed" }

      Result := 1;

end;

Listing 3

TNewCheck3 = class(TCheckBox)

public

  procedure WMGetDlgCode(var Msg: TWMGetDlgCode);

    message wm_GetDlgCode;

end;

...

procedure TNewCheck3.WMGetDlgCode(var Msg: TWMGetDlgCode);

begin

  inherited;

  with Msg do

    { Ensure the result mask has appropriate flag }

    { in to stop keystrokes being "absorbed" }

    Result := Result or dlgc_WantArrows

end;

Listing 4

TForm1 = class(TForm)

...

public

  procedure CMDialogKey(var Msg: TCMDialogKey);

    message cm_DialogKey;

end;

...

procedure TForm1.CMDialogKey(var Msg: TCMDialogKey);

begin

  with Msg do

    if (CharCode in [vk_Up, vk_Down]) and

       (ActiveControl = CheckBox1) then

      { Return 0 to stop the keystrokes being "absorbed" }

      Result := 0

    else

      inherited

end;

